Viable Harvesting of a Renewable Resource

Luc Doyen, Michel De Lara

October 10, 2017

Contents

1 The modelling **1**

2 The viability result **1**

3 The Beverton-Holt case **2**

3.1 Dynamics and equilibrium 2

3.2 The unsustainable case: $\mathbb{V}\text{iab} = \emptyset$ 3

3.3 The sustainable case: $\mathbb{V}\text{iab} \neq \emptyset$ 5

4 Another population dynamics **7**

1 The modelling

Let us consider some regulating agency aiming at the sustainable use and harvesting of a renewable resource. The biomass of this resource at time t is denoted by $B(t)$ while the harvesting level is $h(t)$. We assume that the natural dynamics is described by some growth function `Biol`. Under exploitation, the following controlled dynamics is obtained for $t \in \mathbb{N}$

$$B(t+1) = \mathtt{Biol}(B(t) - h(t)) \tag{1}$$

with the admissibility constraint

$$0 \leq h(t) \leq B(t)\,. \tag{2}$$

The policy goal is to guarantee at each time a minimal harvesting

$$h_{\text{LIM}} \leq h(t)\,, \tag{3}$$

together with a non extinction level for the resource

$$B_{\text{LIM}} \leq B(t). \tag{4}$$

The combination of constraints (2) and (3) yields the state constraint

$$h_{\text{LIM}} \leq B(t)\,. \tag{5}$$

For the sake of simplicity, we assume that $h_{\text{LIM}} > B_{\text{LIM}}$.

2 The viability result

We use the viability approach to handle the problem. We aim at "showing" and recovering numerically the characterization of the viability kernel described in Result 1. We introduce the following notations.

- The *sustainable yield function* Sust is defined by

$$h = \texttt{Sust}(B) \iff B = \texttt{Biol}(B-h) \text{ and } 0 \leq h \leq B\,. \tag{6}$$

- The *maximum sustainable biomass* B_{MSE} and *maximum sustainable yield* h_{MSE} are defined by

$$h_{\text{MSE}} = \texttt{Sust}(B_{\text{MSE}}) = \max_{B \geq 0} \texttt{Sust}(B)\,. \tag{7}$$

- When $h_{\text{LIM}} \leq h_{\text{MSE}}$, the *viability barrier* B_V is the solution of

$$B_V = \min_{B,\, h_{\text{LIM}} = \texttt{Sust}(B)} B\,. \tag{8}$$

Result 1 *Assume that* Biol *is an increasing continuous function on* $\mathbb{R}_+$*. The viability kernel is characterized by*

$$\mathbb{V}\text{iab} = \begin{cases} [B_V, +\infty[& \textit{if } h_{\text{LIM}} \leq h_{\text{MSE}} \\ \emptyset & \textit{otherwise.} \end{cases} \tag{9}$$

Viable controls are those h *(depending on* B*) such that*

$$\texttt{Biol}(B-h) \in [B_V, +\infty[\;\; \textit{and} \;\; 0 \leq h \leq B\,. \tag{10}$$

3 The Beverton-Holt case

3.1 Dynamics and equilibrium

We consider the natural evolution governed by

$$\texttt{Biol}(B) = \frac{RB}{1+bB}\,. \tag{11}$$

For numerical computations and simulations, we consider the case of the Pacific yellowfin tuna, with intrinsic growth $R = 2.25$ metric tons per year and carrying capacity $K = 250\,000$ metric tons. Since the carrying capacity solves $\texttt{Biol}(K) = K$, we obtain

$$R = 2.25 \text{ metric tons per year and } b = \frac{R-1}{K} = 5\;10^{-6} \text{ (metric tons per year)}^{-1}\,. \tag{12}$$

Question 1

1. *Define a Scilab function for the dynamics* `Biol` *in* (11)*, and a Scilab function for the equilibrium harvesting* $h = \texttt{Sust}(B)$ *defined in (6).*

2. *Plot the sustainable yield function* $B \mapsto \texttt{Sust}(B)$*.*

3. *Compute the maximum sustainable population equilibrium* B_{MSE} *and the maximum sustainable yield* h_{MSE}*.*

```
// Population dynamics parameters
R_tuna=2.25;
R=R_tuna;
K_tuna=250000;// carrying capacity
K=K_tuna;

// BEVERTON-HOLT DYNAMICS
b=(R-1)/K;

function y=dynamics(B) y=R*B ./(1+b*B);endfunction

function h=SY(B) h=B-(B ./(R-b*B));endfunction

B_MSE=(R-sqrt(R))/b;
disp('The maximum sustainable population equilibrium is '+string(B_MSE)+ ...
     ' metric tons (MT)');

h_MSE=SY(B_MSE);
disp('The maximum sustainable yield is '+string(h_MSE)+' metric tons (MT)');

xset("window",10);xbasc();
// xbasc() is no longer valid with scilab, but works with scicos
abcisse_B=linspace(0,K,20);
plot2d(abcisse_B,SY(abcisse_B))
xtitle('Sustainable yield associated to Beverton-Holt dynamics','biomass (MT)', ...
       'yield (MT)');
```

3.2 The unsustainable case: $\mathbb{V}\mathrm{iab} = \emptyset$

Question 2

1. *Choose a guaranteed harvesting* h_{LIM} *strictly larger than* $\texttt{Sust}(B_{\mathrm{MSE}})$*.*

2. *For several initial conditions* B_0*, compute different trajectories for the smallest admissible harvesting, namely* $h(t) = h_{\mathrm{LIM}}$*.*

3. *What happens with respect to viability constraint (5)?*

4. *Show that the situation is more catastrophic with sequences of harvesting $h(t)$ larger than the guaranteed one h_{LIM}. Try for instance admissible policies:*
   - $h(t,B) = B$;
   - $h(t,B) = \alpha h_{\text{LIM}} + (1-\alpha)B$ *with* $\alpha \in [0,1]$ *(see Figure 1(a));*
   - $h(t,B) = \alpha(t) h_{\text{LIM}} + (1-\alpha(t))B$, *where* $(\alpha(t))_{t\in\mathbb{N}}$ *is an i.i.d. sequence of uniform random variables in* $[0,1]$ *(see Figure 1(b)).*

```
// number of random initial conditions
nbsimul=20;
// The unsustainable case:
h_lim=(1+rand()/5)*h_MSE

Horizon=30;
years=1:Horizon;
yearss=[years,years($)+1];
traj_B=[];

xset("window",11);xbasc();

for i=1:nbsimul do
  traj_B(1)=K*rand(1);
  for t=years do
    traj_B(t+1)=max(0,dynamics(traj_B(t)-h_lim));
    // max is no longer valid with scilab (use maxi), but works with scicos
  end,
  plot2d(yearss,traj_B);
end,
plot2d(yearss,h_lim*ones(yearss))
xtitle('Population trajectories violating the constraint','years (t)', ...
       'biomass B (metric tons)')
legends('guaranteed harvesting',1,'ur')

///////////////////////////////////////////////////

function h=policy(t,B,alpha)
  h=max(h_lim,alpha*h_lim+(1-alpha)*B);
endfunction

Horizon=8;
years=1:Horizon;
yearss=[years,years($)+1];
traj_B=[];
```



(a) Population viability constraint violated (stationary harvesting)

(b) Population viability constraint violated (random harvesting)

Figure 1: Biomass trajectories violating the viability state constraint (5) (horizontal line at h_{LIM})

```
traj_h=[];
alpha=rand(years);

xset("window",12);xbasc();

for i=1:nbsimul do
  traj_B(1)=K*rand(1);
  for t=years do
    traj_h(t)=policy(t,traj_B(t),alpha(t));
    traj_B(t+1)=max(0,dynamics(traj_B(t)-traj_h(t)));
  end,
  plot2d(yearss,traj_B);
end,
plot2d(yearss,h_lim*ones(yearss))
xtitle('Population trajectories violating the constraint','years (t)', ...
       'biomass B (metric tons)')
legends('guaranteed harvesting',1,'ur')
```

3.3 The sustainable case: $\mathbb{V}\text{iab} \neq \emptyset$

Question 3

1. *Now choose a guaranteed harvesting h_{LIM} strictly smaller than $h_{\text{MSE}} = \texttt{Sust}(B_{\text{MSE}})$.*
2. *Compute the viability barrier B_V.*

3. *Prove that the viable policies are those $h(t,B)$ which lie within the set $[h_{\text{LIM}}, h^\sharp(B)]$ where*
$$h^\sharp(B) = B - \frac{B_V}{R - bB_V} . \qquad (13)$$

4. *For different initial conditions B_0, compute different trajectories for a harvesting policy of the form*
$$h(t,B) = \alpha(t) h_{\text{LIM}} + (1-\alpha(t)) h^\sharp(B) . \qquad (14)$$

5. *What happens with respect to viability constraint (5)? See Figure 2.*

```
// The sustainable case
h_lim=(1-rand()/5)*h_MSE
///////////////////////////////////////////////////
// numerical estimation of the viability barrier
function residual=viabbarrier(B)
  residual=SY(B)-h_lim
endfunction

B_V=fsolve(0,viabbarrier)
///////////////////////////////////////////////////
function h=h_max(B)
  h=B-(B_V/(R-b*B_V))
endfunction

function h=viab(t,B,alpha)
  h=max(h_lim,alpha*h_lim+(1-alpha)*h_max(B));
endfunction

Horizon=10;
years=1:Horizon;
yearss=[years,years($)+1];

xset("window",21);xbasc();
xtitle('Population trajectories ','years (t)','biomass B (metric tons)')
traj_B=[];

xset("window",22);xbasc();
xtitle("Harvesting trajectories satisfying the constraint",'years (t)', ...
       'harvesting h (metric tons)')
traj_h=[];

alpha=rand(years);
```



(a) Population viability constraint satisfied or violated

(b) Minimal guaranteed harvesting viability constraint satisfied

Figure 2: Population trajectories violating or satisfying the viability state constraint (5), depending whether the original biomass is lower or greater than the viability barrier (horizontal line at B_V). Harvest trajectories satisfying the minimal harvesting viability constraint (3) (horizontal line at h_{LIM}).

```
for i=1:nbsimul do
  traj_B(1)=K*rand(1);
  for t=years do
    traj_h(t)=viab(t,traj_B(t),alpha(t));
    traj_B(t+1)=max(0,dynamics(traj_B(t)-traj_h(t)));
  end,
  xset("window",21);plot2d(yearss,traj_B);
  plot2d(yearss,B_V*ones(yearss));
  legends('viability barrier',1,'ur')
  //
  xset("window",22);plot2d(years,traj_h);
  plot2d(years,h_lim*ones(years));
  legends('guaranteed harvesting',1,'ur')
  plot2d(years,0*ones(years));
  // for the scale
end
```

4 Another population dynamics

The natural evolution is now governed by

$$\texttt{Biol}(B) = \sqrt{KB}. \tag{15}$$

```
sqrt(max(0,K*B))
```

Question 4 *Do the same as in the previous Beverton-Holt case.*